# $Id: PKGBUILD 68298 2012-03-21 10:33:42Z pschmitz $
# Maintainer: Ronald van Haren <ronald.archlinux.org>
# Contributor: JJDaNiMoTh <jjdanimoth@gmail.com>
# Contributor: nesl247 <nesl247@gmail.com>

pkgname=compiz-fusion-plugins-extra
pkgver=0.8.8
pkgrel=2
pkgdesc="Compiz Fusion Extra plugins"
arch=('i686' 'x86_64')
url="http://www.compiz.org"
license=('GPL')
depends=('compiz-core' 'compiz-bcop' 'compiz-fusion-plugins-main' 'libnotify')
makedepends=('intltool' 'pkg-config' 'gettext' 'gconf')
groups=('compiz-fusion' 'compiz-fusion-gtk' 'compiz-fusion-kde')
conflicts=('compiz-fusion-plugins-extra-git')
options=('!libtool')
source=(http://releases.compiz-fusion.org/${pkgver}/compiz-plugins-extra-${pkgver}.tar.bz2)
sha1sums=('ce592993696c0b5d4e0bb13171897b4c25b3844e')

build() {
  cd "${srcdir}/compiz-plugins-extra-${pkgver}"

  sed -i 's/if text "x$use/if test "x$use/' configure
  sed -i 's/iconUri, NULL);/iconUri);/' src/notification/notification.c
  ./configure --prefix=/usr

  make
}

package() {
  cd "${srcdir}/compiz-plugins-extra-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
